<commit_before>{ stdenv, fetchpatch, fetchgit, fetchFromGitHub }:

let
  # https://github.com/spwhitt/nix-zsh-completions/pull/2
  nix-zsh-completions = fetchFromGitHub {
    owner = "garbas";
    repo = "nix-zsh-completions";
    rev = "9b7d216ec095ccee541ebfa5f04249aa2964d054";
    sha256 = "1pvmfcqdvdi3nc1jm72f54mwf06yrmlq31pqw6b5fczawcz02jrz";
  };
in stdenv.mkDerivation rec {
  rev = "4f19700919c8ebbaf75755fc0d03716d13183f49";
  name = "zsh-prezto-2015-03-03_rev${builtins.substring 0 7 rev}";
  src = fetchgit {
    url = "https://github.com/sorin-ionescu/prezto";
    inherit rev;
    sha256 = "17mql9mb7zbf8q1nvnci60yrmz5bl9q964i8dv4shz8b42861cdg";
    fetchSubmodules = true;
  };
  patches = [
    (fetchpatch {
      url = "https://github.com/sorin-ionescu/prezto/pull/1028.patch";
      sha256 = "0yrj72s1hiaq13374xa82hxdig4s0kvqjn9apkmw0h7kzggxjfn3";
    })
  ];
  buildPhase = ''
    sed -i -e "s|\''${ZDOTDIR:\-\$HOME}/.zpreztorc|/etc/zpreztorc|g" init.zsh
    sed -i -e "s|\''${ZDOTDIR:\-\$HOME}/.zprezto/|$out/|g" init.zsh
    for i in runcoms/*; do
      sed -i -e "s|\''${ZDOTDIR:\-\$HOME}/.zprezto/|$out/|g" $i
    done
    sed -i -e "s|\''${0:h}/cache.zsh|\''${ZDOTDIR:\-\$HOME}/.zfasd_cache|g" modules/fasd/init.zsh
  '';
  installPhase = ''
    mkdir -p $out/modules/nix
    cp ${nix-zsh-completions}/* $out/modules/nix -R
    cp ./* $out/ -R
  '';
  meta = with stdenv.lib; {
    description = "Prezto is the configuration framework for Zsh; it enriches the command line interface environment with sane defaults, aliases, functions, auto completion, and prompt themes.";
    homepage = https://github.com/sorin-ionescu/prezto;
    license = licenses.mit;
    maintainers = with maintainers; [ garbas ];
    platforms = with platforms; unix;
  };
}
<commit_msg>zsh-prezto: updating "4f19700" -> "0crrj2n"<commit_after>{ stdenv, fetchpatch, fetchgit, fetchFromGitHub }:

stdenv.mkDerivation rec {
  name = "zsh-prezto-2017-12-03";
  src = fetchgit {
    url = "https://github.com/sorin-ionescu/prezto";
    rev = "029414581e54f5b63156f81acd0d377e8eb78883";
    sha256 = "0crrj2nq0wcv5in8qimnkca2an760aqald13vq09s5kbwwc9rs1f";
    fetchSubmodules = true;
  };
  buildPhase = ''
    sed -i -e "s|\''${ZDOTDIR:\-\$HOME}/.zpreztorc|/etc/zpreztorc|g" init.zsh
    sed -i -e "s|\''${ZDOTDIR:\-\$HOME}/.zprezto/|$out/|g" init.zsh
    for i in runcoms/*; do
      sed -i -e "s|\''${ZDOTDIR:\-\$HOME}/.zprezto/|$out/|g" $i
    done
    sed -i -e "s|\''${0:h}/cache.zsh|\''${ZDOTDIR:\-\$HOME}/.zfasd_cache|g" modules/fasd/init.zsh
  '';
  installPhase = ''
    mkdir -p $out
    cp ./* $out/ -R
  '';
  meta = with stdenv.lib; {
    description = "Prezto is the configuration framework for Zsh; it enriches the command line interface environment with sane defaults, aliases, functions, auto completion, and prompt themes.";
    homepage = https://github.com/sorin-ionescu/prezto;
    license = licenses.mit;
    maintainers = with maintainers; [ garbas ];
    platforms = with platforms; unix;
  };
}

{'fraction_non_alphanumeric': 0.16605899900563473, 'fraction_numerical': 0.06463374212794167, 'mean_word_length': 3.3299856527977045, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 6, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '7569673', 'n_tokens_mistral': 1371, 'n_tokens_neox': 1275, 'n_words': 217}